



ГОТОВИМСЯ К ОЛИМПИАДАМ  
ПО ИНФОРМАТИКЕ

*Буздалов Максим Викторович,  
Ульянцев Владимир Игоревич,  
Царев Федор Николаевич*

## ЗАДАЧА «ДВЕ ДУГИ»

Этой статьей мы продолжаем цикл публикаций олимпиадных задач для школьников по информатике и программированию с разборами. Решение таких задач и изучение разборов поможет вам лучше подготовиться к олимпиадам по информатике.

В этой статье рассматривается задача «Две дуги», которая предлагалась в третьей Интернет-олимпиаде сезона 2008–2009 гг. Интернет-олимпиады по информатике проводятся Санкт-Петербургским государственным университетом информационных технологий, механики и оптики в двух номинациях – базовой и усложненной. Базовая номинация рассчитана на начинающих участников олимпиад, поэтому в ней предлагаются более простые задачи, а в усложненной номинации предла-

гаются задачи уровня городских и всероссийских командных олимпиад по программированию. Сайт этих олимпиад находится по адресу <http://neerc.ifmo.ru/school/io>.

### УСЛОВИЕ ЗАДАЧИ

В этой задаче речь пойдет о дугах одной и той же окружности.

Каждая дуга будет задаваться градусными мерами ее двух концов. Эти градусные меры являются целыми числами от  $0^\circ$  до  $359^\circ$  включительно и измеряются против часовой стрелки от луча, направленного вправо. Так как в общем случае существуют две дуги с заданными концами, мы будем считать, что дуга проходит из первой точки в сторону возрастания угла (с учетом перехода через отметку  $0^\circ$ ).


На рис. 1 изображены дуги, задаваемые соответственно как  $(0; 60)$ ,  $(180; 30)$ ,  $(210; 270)$ .


В случае совпадения концов дуг условимся считать, что таким образом задана дуга, состоящая из одной точки.

Заданы две дуги окружности. Необходимо проверить, имеют ли они хотя бы одну общую точку.

### Формат входного файла

В первой строке входного файла находятся два целых числа  $S_1, F_1$  – описание





**Рис. 1.** Примеры дуг, задаваемых как  $(0; 60)$ ,  $(180; 30)$  и  $(210; 270)$

первой дуги. Во второй строке находится описание второй дуги в том же формате.

Все числа во входном файле неотрицательны и строго меньше 360.

#### Формат выходного файла

В выходной файл выведите «YES», если дуги имеют хотя бы одну общую точку, или «NO», если не имеют общих точек.

#### *Примеры входных и выходных данных*

<b>twoarcs.in</b>	<b>twoarcs.out</b>
0 60	NO
210 270	
210 270	YES
180 30	
180 30	YES
0 60	

### РАЗБОР ЗАДАЧИ

Сначала решим вспомогательную задачу – для дуги и точки на окружности определим, лежит ли точка на дуге.

Пусть дуга начинается в точке с градусной мерой  $S$  и заканчивается в точке с градусной мерой  $F$  ( $1 \leq S, F < 360$ ), а точка, принадлежность которой дуге необходимо определить, имеет градусную меру  $X$ . Рассмотрим два случая:  $S \leq F$  и  $S > F$ .

В первом случае дуга содержит только точки с градусной мерой из отрезка  $[S; F]$ , поэтому данная точка принадлежит дуге тогда и только тогда, когда  $S \leq X \leq F$ .

Во втором случае дуга содержит точку с градусной мерой 0 – при переходе через эту точку значение угла скачкообразно уменьшается. Так как во всех ос-

тальных точках окружности такого не происходит, то дуга содержит те и только те точки, градусные меры которых принадлежат объединению интервалов  $[S; 360]$  и  $[0; F]$ . Поэтому точка принадлежит дуге тогда и только тогда, когда  $S \leq X$  или  $X \leq F$ .

Отсюда следует первый способ решения. Так как концы имеют целочисленные градусные меры, то для проверки наличия у них общих точек достаточно проверить также только точки с целочисленными градусными мерами (детальное доказательство остается читателю в качестве упражнения). В листинге 1 приведена программная реализация этого подхода на языке Pascal.

Эта программа работает за линейное время относительно числа целочисленных значений градусных мер. Это число равно 360, поэтому решение укладывается в ограничения по времени.

Можно заметить, что дуги имеют хотя бы одну общую точку тогда и только тогда, когда верно хотя бы одно из четырех утверждений:

- первая дуга содержит начало второй дуги;
- первая дуга содержит конец второй дуги;
- вторая дуга содержит начало первой дуги;
- вторая дуга содержит конец первой дуги.

На основании этого можно реализовать второй способ решения, работающий за константное время. Программная реализация приведена в листинге 2.

**Листинг 1.** Первый способ решения

```
uses
  sysutils;
var
  s1, f1, s2, f2, i: integer;
  ans, v1, v2: boolean;
begin
  reset(input, 'twoarcs.in');
  rewrite(output, 'twoarcs.out');
  readln(s1, f1);
  readln(s2, f2);
  ans := false;
  for i := 0 to 359 do begin
    if s1 <= f1 then begin
      v1 := (s1 <= i) and (i <= f1);
    end else begin
      v1 := (s1 <= i) or (i <= f1);
    end;
    if s2 <= f2 then begin
      v2 := (s2 <= i) and (i <= f2);
    end else begin
      v2 := (s2 <= i) or (i <= f2);
    end;
    ans := ans or (v1 and v2);
  end;
  if ans then begin
    writeln('YES');
  end else begin
    writeln('NO');
  end;
end.
```

**Листинг 2.** Второй способ решения

```
uses
  sysutils;
var
  s1, f1, s2, f2, i: integer;
  v1, v2, v3, v4: boolean;
begin
  reset(input, 'twoarcs.in');
  rewrite(output, 'twoarcs.out');
  readln(s1, f1);
  readln(s2, f2);
  if s1 <= f1 then begin
    v1 := (s1 <= s2) and (s2 <= f1);
    v2 := (s1 <= f2) and (f2 <= f1);
  end else begin
    v1 := (s1 <= s2) or (s2 <= f1);
    v2 := (s1 <= f2) or (f2 <= f1);
  end;
  if s2 <= f2 then begin
    v3 := (s2 <= s1) and (s1 <= f2);
    v4 := (s2 <= f1) and (f1 <= f2);
  end else begin
    v3 := (s2 <= s1) or (s1 <= f2);
    v4 := (s2 <= f1) or (f1 <= f2);
  end;
  if v1 or v2 or v3 or v4 then begin
    writeln('YES');
  end else begin
    writeln('NO');
  end;
end.
```

Отметим, что второе решение можно модифицировать так, чтобы оно работало для вещественных значений градусных мер концов заданных дуг. Для первого решения такая модификация затруднительна.

**Члены жюри Интернет-олимпиад по информатике базового уровня:**

**Буздалов Максим Викторович,**  
аспирант кафедры «Компьютерные технологии» (КТ) СПбГУ ИТМО,  
чемпион мира по программированию среди студентов 2009 года,

**Ульянцев Владимир Игоревич,**  
студент пятого курса кафедры КТ  
СПбГУ ИТМО, член жюри ВКОШП,

**Царёв Федор Николаевич,**  
аспирант кафедры КТ СПбГУ  
ИТМО, чемпион мира по  
программированию среди  
студентов 2008 года.



Наши авторы, 2011.  
Our authors, 2011.